3

CONFIGURING YOUR PROJECT WITH AUTOCONF

Come my friends, 'Tis not too late to seek a newer world. —Alfred, Lord Tennyson, "Ulysses"



Because Automake and Libtool are essentially add-on components to the original Autoconf framework, it's useful to spend

some time focusing on using Autoconf without Automake and Libtool. This will provide a fair amount of insight into how Autoconf operates by exposing aspects of the tool that are often hidden by Automake.

Before Automake came along, Autoconf was used alone. In fact, many legacy open source projects never made the transition from Autoconf to the full GNU Autotools suite. As a result, it's not unusual to find a file called *configure.in* (the original Autoconf naming convention) as well as handwritten *Makefile.in* templates in older open source projects.

In this chapter, I'll show you how to add an Autoconf build system to an existing project. I'll spend most of this chapter talking about the fundamental features of Autoconf, and in Chapter 4, I'll go into much more detail about how some of the more complex Autoconf macros work and how to properly use them. Throughout this process, we'll continue using the Jupiter project as our example.

Autoconf Configuration Scripts

The input to the autoconf program is shell script sprinkled with macro calls. The input stream must also include the definitions of all referenced macros—both those that Autoconf provides and those that you write yourself.

The macro language used in Autoconf is called M4. (The name means M, *plus 4 more letters*, or the word *Macro*.¹) The m4 utility is a general-purpose macro language processor originally written by Brian Kernighan and Dennis Ritchie in 1977.

While you may not be familiar with it, you can find some form of M4 on every Unix and Linux variant (as well as other systems) in use today. The prolific nature of this tool is the main reason it's used by Autoconf, as the original design goals of Autoconf stated that it should be able to run on all systems without the addition of complex tool chains and utility sets.²

Autoconf depends on the existence of relatively few tools: a Bourne shell, M4, and a Perl interpreter. The configuration scripts and makefiles it generates rely on the existence of a different set of tools, including a Bourne shell, grep, 1s, and sed or awk.³

NOTE Do not confuse the requirements of the Autotools with the requirements of the scripts and makefiles they generate. The Autotools are maintainer tools, while the resulting scripts and makefiles are end-user tools. We can reasonably expect a higher level of installed functionality on development systems than we can on end-user systems.

The configuration script ensures that the end user's build environment is configured to properly build your project. This script checks for installed tools, utilities, libraries, and header files, as well as for specific functionality within these resources. What distinguishes Autoconf from other project configuration frameworks is that Autoconf tests also ensure that these resources can be properly consumed by your project. You see, it's not only important that your users have *libxyz.so* and its public header files properly installed on their systems, but also that they have the correct versions of these files. Autoconf is pathological about such tests. It ensures that the end user's environment is in compliance with the project requirements by compiling and linking a small test program for each feature—a quintessential example, if you will, that does what your project source code does on a larger scale.

Can't I just ensure that libxyz.2.1.0.so is installed by searching library paths for the filename? The answer to this question is debatable. There are legitimate situations where libraries and tools get updated quietly. Sometimes, the specific functionality upon which your project relies is added in the form of a security bug fix or enhancement to a library, in which case vendors aren't even required to bump up the version number. But it's often difficult to tell whether you've got version 2.1.0.r1 or version 2.1.0.r2 unless you look at the file size or call a library function to make sure it works as expected.

^{1.} As a point of interest, this naming convention is a fairly common practice in some software engineering domains. For example, the term *internationalization* is often abbreviated *i18n*, for the sake of brevity (or perhaps just because programmers love acronyms).

^{2.} In fact, whatever notoriety M4 may have today is likely due to the widespread use of Autoconf. 3. Autoconf versions 2.62 and later generate configuration scripts that require awk in addition to sed on the end user's system.

However, the most significant reason for not relying on library version numbers is that they do not represent specific marketing releases of a library. As we will discuss in Chapter 7, library version numbers indicate binary interface characteristics on a particular platform. This means that library version numbers for the same feature set can be different from platform to platform, which means that you may not be able to tell—short of compiling and linking against the library—whether or not a particular library has the functionality your project needs.

Finally, there are several important cases where the same functionality is provided by entirely different libraries on different systems. For example, you may find cursor manipulation functionality in *libtermcap* on one system, *libncurses* on another, and *libcurses* on yet another system. But it's not critical that you know about all of these side cases, because your users will tell you when your project won't build on their system because of such a discrepancy.

What can you do when such a bug is reported? You can use the Autoconf AC_SEARCH_LIBS macro to test multiple libraries for the same functionality. Simply add a library to the search list, and you're done. Since this fix is so easy, it's likely the user who noticed the problem will simply send a patch to your *configure.ac* file.

Because Autoconf tests are written in shell script, you have a lot of flexibility as to how the tests operate. You can write a test that merely checks for the existence of a library or utility in the usual locations on your user's system, but this bypasses some of the most significant features of Autoconf. Fortunately, Autoconf provides dozens of macros that conform to Autoconf's feature-testing philosophy. You should carefully study and use the list of available macros, rather than write your own, because they're specifically designed to ensure that the desired functionality is available on the widest variety of systems and platforms.

The Shortest configure.ac File

The simplest possible *configure.ac* file has just two lines, as shown in Listing 3-1.

```
AC_INIT([Jupiter], [1.0])
AC OUTPUT
```

Listing 3-1: The simplest configure.ac file

To those new to Autoconf, these two lines appear to be a couple of function calls, perhaps in the syntax of some obscure programming language. Don't let their appearance throw you—these are M4 macro calls. The macros are defined in files distributed with the *autoconf* package. You can find the definition of AC_INIT, for example, in the *autoconf/general.m4* file in Autoconf's installation directory (usually /*usr/(local/)share/autoconf*). AC_OUTPUT is defined in *autoconf/status.m4*.

Comparing M4 to the C Preprocessor

M4 macros are similar in many ways to the C-preprocessor (CPP) macros defined in C-language source files. The C preprocessor is also a text replacement tool, which isn't surprising: Both M4 and the C preprocessor were designed and written by Kernighan and Ritchie around the same time.

Autoconf uses square brackets around macro parameters as a quoting mechanism. Quotes are necessary only for cases in which the context of the macro call could cause an ambiguity that the macro processor may resolve incorrectly (usually without telling you). We'll discuss M4 quoting in much more detail in Chapter 10. For now, just use square brackets around every argument to ensure that the expected macro expansions are generated.

Like CPP macros, you can define M4 macros to accept a comma-delimited list of arguments enclosed in parentheses. In both utilities, the opening parenthesis must immediately follow the macro name in its definition, with no intervening whitespace. A significant difference, however, is that in M4, the arguments to parameterized macros are optional, and the caller may simply omit them. If no arguments are passed, you can also omit the parentheses. Extra arguments passed to M4 macros are simply ignored. Finally, M4 does not allow intervening whitespace between a macro name and the opening parenthesis in a macro call.

The Nature of M4 Macros

If you've been programming in C for many years, you've no doubt run across a few C-preprocessor macros from the dark regions of the lower realm. I'm talking about those truly evil macros that expand into one or two pages of C code. They should have been written as C functions, but their authors were either overly worried about performance or just got carried away, and now it's your turn to debug and maintain them. But, as any veteran C programmer will tell you, the slight performance gains you get by using a macro where you should have used a function do not justify the trouble you cause maintainers trying to debug your fancy macros. Debugging such macros can be a nightmare because the source code generated by macros is usually inaccessible from within a symbolic debugger.⁴

Writing such complex macros is viewed by M4 programmers as a sort of macro nirvana—the more complex and functional they are, the "cooler" they are. The two Autoconf macros in Listing 3-1 expand into a file containing over 2,200 lines of Bourne-shell script that total more than 60KB in size! But you wouldn't guess this by looking at their definitions. They're both fairly short—only a few dozen lines each. The reason for this apparent disparity is simple: They're written in a modular fashion, each macro expanding several others, which, in turn, expand several others, and so on.

^{4.} A technique I've used in the past for debugging large macros involves manually generating source code using the C preprocessor, and then compiling this generated source. Symbolic debuggers can only work with the source code you provide. By providing source with the macros fully expanded, you enable the debugger to allow you to step through the generated source.

For the same reasons that programmers are taught not to abuse the C preprocessor, the extensive use of M4 causes a fair amount of frustration for those trying to understand Autoconf. That's not to say Autoconf shouldn't use M4 this way; quite the contrary—this is the domain of M4. But there is a school of thought that says M4 was a poor choice for Autoconf because of the problems with macros mentioned above. Fortunately, being able to use Autoconf effectively usually doesn't require a deep understanding of the inner workings of the macros that ship with it.⁵

Executing autoconf

Running autoconf is simple: Just execute it in the same directory as your *configure.ac* file. While I could do this for each example in this chapter, I'm going to use the autoreconf program instead of the autoconf program, because running autoreconf has exactly the same effect as running autoconf, except that autoreconf will also do the right thing when you start adding Automake and Libtool functionality to your build system. That is, it will execute all of the Autotools in the right order based on the contents of your *configure.ac* file.

autoreconf is smart enough to only execute the tools you need, in the order you need them, with the options you want (with one caveat that I'll mention shortly). Therefore, running autoreconf is the recommended method for executing the Autotools tool chain.

Let's start by adding the simple *configure.ac* file from Listing 3-1 to our project directory. The top-level directory currently contains only a *Makefile* and a *src* directory which contains its own *Makefile* and a *main.c* file. Once you've added *configure.ac* to the top-level directory, run autoreconf:

autoreconf	
ls -1p	
utom4te.cache/	
onfigure	
onfigure.ac	
akefile	
rc/	

First, notice that autoreconf operates silently by default. If you want to see something happening, use the -v or --verbose option. If you want autoreconf to execute the Autotools in verbose mode as well, then add -vv to the command line.⁶

Next, notice that autoconf creates a directory called *autom4te.cache*. This is the autom4te cache directory. This cache speeds up access to *configure.ac* during successive executions of utilities in the Autotools tool chain.

^{5.} There are a few exceptions to this rule. Poor documentation can sometimes lead to a misunderstanding about the intended use of some of the published Autoconf macros. This book highlights a few of these situations, but a degree of expertise with M4 is the only way to work your way through most of these problems.

^{6.} You may also pass --verbose --verbose, but this syntax seems a bit . . . verbose to me.

The result of passing *configure.ac* through autoconf is essentially the same file (now called configure), but with all of the macros fully expanded. You're welcome to take a look at configure, but don't be too surprised if you don't immediately understand what you see. The *configure.ac* file has been transformed, through M4 macro expansions, into a text file containing thousands of lines of complex Bourne shell script.

Executing configure

As discussed in "Configuring Your Package" on page 54, the *GNU Coding Standards* indicate that a handwritten configure script should generate another script called config.status, whose job it is to generate files from templates. Unsurprisingly, this is exactly the sort of functionality you'll find in an Autoconf-generated configuration script. This script has two primary tasks:

- Perform requested checks
- Generate and then call config.status

The results of the checks performed by configure are written into config.status in a manner that allows them to be used as replacement text for Autoconf substitution variables in template files (*Makefile.in, config.h.in,* and so on). When you execute configure, it tells you that it's creating config.status. It also creates a log file called *config.log* that has several important attributes. Let's run configure and then see what's new in our project directory.

```
$ ./configure
configure: creating ./config.status
$
$ ls -1p
autom4te.cache/
config.log
config.status
configure
configure.ac
Makefile
src/
$
```

We see that configure has indeed generated both config.status and *config.log*. The *config.log* file contains the following information:

- The command line that was used to invoke configure (very handy!)
- Information about the platform on which configure was executed
- Information about the core tests configure executed
- The line number in configure at which config.status is generated and then called

At this point in the log file, config.status takes over generating log information and adds the following information:

• The command line used to invoke config.status

After config.status generates all the files from their templates, it exits, returning control to configure, which then appends the following information to the log:

- The cache variables config.status used to perform its tasks
- The list of output variables that may be replaced in templates
- The exit code configure returned to the shell

This information is invaluable when debugging a configure script and its associated *configure.ac* file.

Why doesn't configure just execute the code it writes into config.status instead of going to all the trouble of generating a second script, only to immediately call it? There are a few good reasons. First, the operations of performing checks and generating files are conceptually different, and make works best when conceptually different operations are associated with separate make targets. A second reason is that you can execute config.status separately to regenerate output files from their corresponding template files, saving the time required to perform those lengthy checks. Finally, config.status is written to remember the parameters originally used on the configure command line. Thus, when make detects that it needs to update the build system, it can call config.status to re-execute configure, using the command-line options that were originally specified.

Executing config.status

Now that you know how configure works, you might be tempted to execute config.status yourself. This was exactly the intent of the Autoconf designers and the authors of the *GCS*, who originally conceived these design goals. However, a more important reason for separating checks from template processing is that make rules can use config.status to regenerate makefiles from their templates when make determines that a template is newer than its corresponding makefile.

Rather than call configure to perform needless checks (your environment hasn't changed—just your template files), makefile rules should be written to indicate that output files are dependent on their templates. The commands for these rules run config.status, passing the rule's target as a parameter. If, for example, you modify one of your *Makefile.in* templates, make calls config.status to regenerate the corresponding *Makefile*, after which, make re-executes its own original command line—basically restarting itself.⁷

^{7.} This is a built-in feature of GNU make. However, for the sake of portability, Automake generates makefiles that carefully reimplement this functionality as much as possible in make script, rather than relying on the built-in mechanism found in GNU make. The Automake solution isn't quite as comprehensive as GNU make's built-in functionality, but it's the best we can do, under the circumstances.

Listing 3-2 shows the relevant portion of such a *Makefile.in* template, containing the rules needed to regenerate the corresponding *Makefile*.

```
Makefile: Makefile.in config.status
    ./config.status $@
...
```

Listing 3-2: A rule that causes make to regenerate Makefile if its template has changed

A rule with a target named *Makefile* is the trigger here. This rule allows make to regenerate the source makefile from its template if the template changes. It does this *before* executing either the user's specified targets or the default target, if no specific target was given.

The rule in Listing 3-2 indicates that *Makefile* is dependent on config.status as well as *Makefile.in*, because if configure updates config.status, it may generate the makefile differently. Perhaps different command-line options were provided so that configure can now find libraries and header files it couldn't find previously. In this case, Autoconf substitution variables may have different values. Thus, *Makefile* should be regenerated if either *Makefile.in* or config.status is updated.

Since config.status is itself a generated file, it stands to reason that you could write such a rule to regenerate this file when needed. Expanding on the previous example, Listing 3-3 adds the required code to rebuild config.status if configure changes.

```
...
Makefile: Makefile.in config.status
./config.status $@
config.status: configure
./config.status --recheck
...
```

Listing 3-3: A rule to rebuild config.status when configure changes

Since config.status is a dependency of *Makefile*, make will look for a rule whose target is config.status and run its commands if configure is newer than config.status.

Adding Some Real Functionality

I've suggested before that you should call config.status in your makefiles to generate those makefiles from templates. Listing 3-4 shows the code in *configure.ac* that actually makes this happen. It's just a single additional macro call between the two original lines of Listing 3-1.

```
AC_INIT([Jupiter],[1.0])
AC_CONFIG_FILES([Makefile src/Makefile])
AC_OUTPUT
```

Listing 3-4: configure.ac: Using the AC_CONFIG_FILES macro

This code assumes that templates exist for *Makefile* and *src/Makefile*, called *Makefile.in* and *src/Makefile.in*, respectively. These template files look exactly like their *Makefile* counterparts, with one exception: Any text that I want Autoconf to replace is marked as an Autoconf substitution variable, using the @VARIABLE@ syntax.

To create these files, simply rename the existing *Makefiles* to *Makefile.in* in both the top-level and *src* directories. This is a common practice when *autoconfiscating* a project:

```
$ mv Makefile Makefile.in
$ mv src/Makefile src/Makefile.in
$
```

Next, let's add a few Autoconf substitution variables to replace the original default values. At the top of these files, I've also added the Autoconf substitution variable, @configure_input@, after a comment hash mark. Listing 3-5 shows the comment text that is generated in *Makefile*.

```
# Makefile. Generated from Makefile.in by configure.
```

•••

Listing 3-5: Makefile: The text generated from the Autoconf@configure_input@ variable

I've also added the makefile regeneration rules from the previous examples to each of these templates, with slight path differences in each file to account for their different positions relative to config.status and configure in the build directory.

Listings 3-6 and 3-7 highlight in bold the required changes to the final versions of *Makefile* and *src/Makefile* from the end of Chapter 2.

@configure_input@

<pre># Package-speci package version tarname distdir</pre>	= = =	substitution variables @PACKAGE_NAME@ @PACKAGE_VERSION@ @PACKAGE_TARNAME@ \$(tarname)-\$(version)
<pre># Prefix-specif: prefix exec_prefix bindir</pre>	=	substitution variables @prefix@ @exec_prefix@ @bindir@

• • •

```
$(distdir): FORCE
        mkdir -p $(distdir)/src
        cp configure.ac $(distdir)
        cp configure $(distdir)
        cp Makefile.in $(distdir)
        cp src/Makefile.in $(distdir)/src
        cp src/main.c $(distdir)/src
distcheck: $(distdir).tar.gz
        gzip -cd $(distdir).tar.gz | tar xvf -
        cd $(distdir) && ./configure
        cd $(distdir) && $(MAKE) all
        cd $(distdir) && $(MAKE) check
        cd $(distdir) && $(MAKE) DESTDIR=$${PWD}/ inst install
        cd $(distdir) && $(MAKE) DESTDIR=$${PWD}/ inst uninstall
        @remaining="`find $${PWD}/$(distdir)/ inst -type f | wc -l`"; \
        if test "$${remaining}" -ne 0; then \
          echo "*** $${remaining} file(s) remaining in stage directory!"; \
          exit 1; \
        fi
        cd $(distdir) && $(MAKE) clean
        rm -rf $(distdir)
        @echo "*** Package $(distdir).tar.gz is ready for distribution."
Makefile: Makefile.in config.status
        ./config.status $@
config.status: configure
        ./config.status --recheck
```

Listing 3-6: Makefile.in: Required modifications to Makefile from the end of Chapter 2

```
# @configure_input@
```

```
# Package-specific substitution variables
               = @PACKAGE_NAME@
package
               = @PACKAGE VERSION@
version
               = @PACKAGE_TARNAME@
tarname
distdir
               = $(tarname)-$(version)
# Prefix-specific substitution variables
               = @prefix@
prefix
exec prefix
               = @exec prefix@
bindir
               = @bindir@
. . .
Makefile: Makefile.in ../config.status
       cd .. && ./config.status src/$@
```

```
../config.status: ../configure
    cd .. && ./config.status --recheck
```

Listing 3-7: src/Makefile.in: Required modifications to src/Makefile from the end of Chapter 2

I've removed the export statements from the top-level *Makefile.in* and added a copy of all of the make variables (originally only in the top-level *Makefile*) into *src/Makefile.in*. Since config.status generates both of these files, I can reap excellent benefits by substituting values for these variables directly into both files. The primary advantage of doing this is that I can now run make in any subdirectory without worrying about uninitialized variables that would originally have been passed down by a higher-level makefile.

Since Autoconf generates entire values for these make variables, you may be tempted to clean things up a bit by removing the variables and just substituting @prefix@ where we currently use \$(prefix) throughout the files. There are a few good reasons for keeping the make variables. First and foremost, we'll retain the original benefits of the make variables; our end users can continue to substitute their own values on the make command line. (Even though Autoconf places default values in these variables, users may wish to override them.) Second, for variables such as \$(distdir), whose values are comprised of multiple variable references, it's simply cleaner to build the name in one place and use it everywhere else through a single variable.

I've also changed the commands in the distribution targets a bit. Rather than distribute the makefiles, I now need to distribute the *Makefile.in* templates, as well as the new configure script and the *configure.ac* file.⁸

Finally, I modified the distcheck target's commands to run the configure script before running make.

Generating Files from Templates

Note that you can use AC_CONFIG_FILES to generate *any* text file from a file of the same name with an *.in* extension, found in the same directory. The *.in* extension is the default template naming pattern for AC_CONFIG_FILES, but you can override this default behavior. I'll get into the details shortly.

Autoconf generates sed or awk expressions into the resulting configure script, which then copies them into config.status. The config.status script uses these expressions to perform string replacement in the input template files.

Both sed and awk are text-processing tools that operate on file streams. The advantage of a stream editor (the name *sed* is a contraction of the phrase *stream editor*) is that it replaces text patterns in a byte stream. Thus, both sed and awk can operate on huge files because they don't need to load the entire input file into memory in order to process it. Autoconf builds the expression list that config.status passes to sed or awk from a list of variables defined by

^{8.} Distributing *configure.ac* is not merely an act of kindness—it could also be considered a requirement of GNU source licenses, since *configure.ac* is very literally the source code for configure.

various macros, many of which I'll cover in greater detail later in this chapter. It's important to understand that Autoconf substitution variables are the *only* items replaced in a template file while generating output files.

At this point, with very little effort, I've created a basic *configure.ac* file. I can now execute autoreconf, followed by configure and then make, in order to build the Jupiter project. This simple, three-line *configure.ac* file generates a configure script that is fully functional, according to the definition of a proper configuration script defined by the *GCS*.

The resulting configuration script runs various system checks and generates a config.status script that can replace a fair number of substitution variables in a set of specified template files in this build system. That's a lot of functionality in just three lines of code.

Adding VPATH Build Functionality

At the end of Chapter 2, I mentioned that I hadn't yet covered an important concept—that of VPATH builds. A *VPATH build* is a way of using a makefile construct (VPATH) to configure and build a project in a directory other than the source directory. This is important if you need to perform any of the following tasks:

- Maintain a separate debug configuration
- Test different configurations side by side
- Keep a clean source directory for patch diffs after local modifications
- Build from a read-only source directory

The VPATH keyword is short for *virtual search path*. A VPATH statement contains a colon-separated list of places to look for relative-path dependencies when they can't be found relative to the current directory. In other words, when make can't find a prerequisite file relative to the current directory, it searches for that file successively in each of the paths in the VPATH statement.

Adding remote build functionality to an existing makefile using VPATH is very simple. Listing 3-8 shows an example of using a VPATH statement in a makefile.

program: src/main.c \$(CC)

Listing 3-8: An example of using VPATH in a makefile

In this (contrived) example, if make can't find *src/main.c* in the current directory while processing the rule, it will look for *some/path/src/main.c*, and then for *some/other/path/src/main.c*, and finally for *yet/another/path/src/main.c* before giving up with an error message about not knowing how to make *src/main.c*.

With just a few simple modifications, we can completely support remote builds in Jupiter. Listings 3-9 and 3-10 illustrate the necessary changes to the project's two makefiles.

```
...
# VPATH-specific substitution variables
srcdir = @srcdir@
VPATH = @srcdir@
...
$(distdir): FORCE
    mkdir -p $(distdir)/src
    cp $(srcdir)/configure.ac $(distdir)
    cp $(srcdir)/configure $(distdir)
    cp $(srcdir)/makefile.in $(distdir)
    cp $(srcdir)/src/Makefile.in $(distdir)/src
    cp $(srcdir)/src/main.c $(distdir)/src
...
```

Listing 3-9: Makefile.in: Adding VPATH build capabilities to the top-level makefile

```
...
# VPATH-related substitution variables
srcdir = @srcdir@
VPATH = @srcdir@
...
```

Listing 3-10: src/Makefile.in: Adding VPATH build capabilities to the lower-level makefile

That's it. Really. When config.status generates a file, it replaces an Autoconf substitution variable called @srcdir@ with the relative path to the template's source directory. The value substituted for @srcdir@ in a given *Makefile* within the build directory structure is the relative path to the directory containing the corresponding *Makefile.in* template in the source directory structure. The concept here is that for each *Makefile* in the remote build directory, VPATH provides a relative path to the directory containing the source code for that build directory.

The changes required for supporting remote builds in your build system are summarized as follows:

- Set a make variable, srcdir, to the @srcdir@ substitution variable.
- Set the VPATH variable to @srcdir@.
- Prefix all file dependencies used *in commands* with \$(srcdir)/.

```
NOTE Don't use $(srcdir) in the VPATH statement itself, because some older versions of make won't substitute variable references within the VPATH statement.
```

If the source directory is the same as the build directory, the @srcdir@ substitution variable degenerates to a dot (.). That means all of these \$(srcdir)/ prefixes simply degenerate to ./, which is harmless.⁹

^{9.} This is not strictly true for non-GNU implementations of make. GNU make is smart enough to know that *file* and *./file* refer to the same filesystem object. However, non-GNU implementations of make aren't always quite so intelligent, so you should be careful to refer to a filesystem object using the same notation for each reference in your *Makefile.in* templates.

A quick example is the easiest way to show you how this works. Now that Jupiter is fully functional with respect to remote builds, let's give it a try. Start in the Jupiter project directory, create a subdirectory called *build*, and then change into that directory. Execute the configure script using a relative path, and then list the current directory contents:

```
$ mkdir build
$ cd build
$ ../configure
configure: creating ./config.status
config.status: creating Makefile
config.status: creating src/Makefile
$
$ ls -1p
config.log
config.status
Makefile
src/
$
$ ls -1p src
Makefile
$
```

The entire build system has been constructed by configure and config.status within the *build* subdirectory. Enter make to build the project from within the *build* directory:

```
$ make
cd src && make all
make[1]: Entering directory '../prj/jupiter/build'
gcc -g -02 -0 jupiter .././src/main.c
make[1]: Leaving directory '../prj/jupiter/build'
$
$ ls -1p src
jupiter
Makefile
$
```

No matter where you are, if you can access the project directory using either a relative or an absolute path, you can do a remote build from that location. This is just one more thing that Autoconf does for you in Autoconfgenerated configuration scripts. Imagine managing proper relative paths to source directories in your own hand-coded configuration scripts!

Let's Take a Breather

So far, I've shown you a nearly complete build system that includes almost all of the features outlined in the *GCS*. The features of Jupiter's build system are all fairly self-contained and reasonably simple to understand. The most difficult feature to implement by hand is the configuration script. In fact, writing

a configuration script by hand is so labor intensive, compared to the simplicity of using Autoconf, that I just skipped the hand-coded version entirely in Chapter 2.

Although using Autoconf like I've used it here is quite easy, most people don't create their build systems in the manner I've shown you. Instead, they try to copy the build system of another project, and tweak it to make it work in their own project. Later, when they start a new project, they do the same thing again. This can cause trouble because the code they're copying was never meant to be used the way they're now trying to use it.

I've seen projects in which the *configure.ac* file contained junk that had nothing to do with the project to which it belonged. These leftover bits came from some legacy project, but the maintainer didn't know enough about Autoconf to properly remove all the extraneous text. With the Autotools, it's generally better to start small and add what you need than to start with a copy of *configure.ac* from another full-featured build system, and try to pare it down to size or otherwise modify it to work with a new project.

I'm sure you're feeling like there's a lot more to learn about Autoconf, and you're right. We'll spend the majority of this chapter examining the most important Autoconf macros and how they're used in the context of the Jupiter project. But first, let's go back and see if we might be able to simplify the Autoconf startup process even more by using another utility that comes with the *autoconf* package.

An Even Quicker Start with autoscan

The easiest way to create a (mostly) complete *configure.ac* file is to run the autoscan utility, which is part of the *autoconf* package. This utility examines the contents of a project directory and generates the basis for a *configure.ac* file (which autoscan names *configure.scan*) using existing makefiles and source files.

Let's see how well autoscan does on the Jupiter project. First, I'll clean up the droppings from my earlier experiments, and then run autoscan in the *jupiter* directory. Note that I'm *not* deleting my original *configure.ac* file—I'll just let autoscan tell me how to improve it. In less than a second, I have a few new files in the top-level directory:

```
$ rm -rf autom4te.cache build
$ rm configure config.* Makefile src/Makefile src/jupiter
$ ls -1p
configure.ac
Makefile.in
src/
$
$ autoscan
• configure.ac: warning: missing AC_CHECK_HEADERS([stdlib.h]) wanted by: src/main.c:2
configure.ac: warning: missing AC_PREREQ wanted by: autoscan
configure.ac: warning: missing AC_PROG_CC wanted by: src/main.c
configure.ac: warning: missing AC_PROG_INSTALL wanted by: Makefile.in:18
$
```

\$ ls -1p autom4te.cache/ autoscan.log configure.ac configure.scan Makefile.in src/ \$

The autoscan utility examines the project directory hierarchy and creates two files called *configure.scan* and *autoscan.log*. The project may or may not already be instrumented for Autotools—it doesn't really matter, because autoscan is decidedly non-destructive. It will never alter any existing files in a project.

The autoscan utility generates a warning message for each problem it discovers in an existing *configure.ac* file. In this example, autoscan noticed that *configure.ac* should be using the Autoconf-provided AC_CHECK_HEADERS, AC_PREREQ, AC_PROG_CC, and AC_PROG_INSTALL macros. It made these assumptions based on information gleaned from the existing *Makefile.in* templates and from the C-language source files, as you can see by the comments after the warning statements beginning at **①**. You can always see these messages (in even greater detail) by examining the *autoscan.log* file.

NOTE The notices you receive from autoscan and the contents of your configure.ac file may differ slightly from mine, depending on the version of Autoconf you have installed. I have version 2.64 of GNU Autoconf installed on my system (the latest, as of this writing). If your version of autoscan is older (or newer), you may see some minor differences.

Looking at the generated *configure.scan* file, I note that autoscan has added more text to this file than was in my original *configure.ac* file. After looking it over to ensure that I understand everything, I see that it's probably easiest for me to overwrite *configure.ac* with *configure.scan* and then change the few bits of information that are specific to Jupiter:

```
$ mv configure.scan configure.ac
$ cat configure.ac
# -*- Autoconf -*-
# Process this file with autoconf to produce a configure script.
AC_PREREQ([2.64])
AC_INIT([FULL-PACKAGE-NAME], [VERSION], [BUG-REPORT-ADDRESS])
AC_CONFIG_SRCDIR([src/main.c])
AC_CONFIG_HEADERS([config.h])
# Checks for programs.
AC_PROG_CC
AC_PROG_INSTALL
# Checks for libraries.
```

```
# Checks for header files.
AC_CHECK_HEADERS([stdlib.h])
# Checks for typedefs, structures, and compiler characteristics.
# Checks for library functions.
AC_CONFIG_FILES([Makefile
src/Makefile])
AC_OUTPUT
$
```

My first modification involves changing the AC_INIT macro parameters for Jupiter, as illustrated in Listing 3-11.

```
# -*- Autoconf -*-
# Process this file with autoconf to produce a configure script.
AC_PREREQ([2.64])
AC_INIT([Jupiter], [1.0], [jupiter-bugs@example.org])
AC_CONFIG_SRCDIR([src/main.c])
AC_CONFIG_HEADERS([config.h])
...
```

Listing 3-11: configure.ac: Tweaking the AC_INIT macro generated by autoscan

The autoscan utility does a lot of the work for you. The *GNU Autoconf Manual*¹⁰ states that you should modify this file to meet the needs of your project before you use it, but there are only a few key issues to worry about (besides those related to AC_INIT). I'll cover each of these issues in turn, but first, let's take care of a few administrative details.

The Proverbial autogen.sh Script

Before autoreconf came along, maintainers passed around a short shell script, often named autogen.sh or bootstrap.sh, which would run all of the Autotools required for their projects in the proper order. The autogen.sh script can be fairly sophisticated, but to solve the problem of the missing install-sh script (see "Missing Required Files in Autoconf" on page 74), I'll just add a simple temporary autogen.sh script to the project root directory, as shown in Listing 3-12.

```
#!/bin/sh
autoreconf --install
automake --add-missing --copy >/dev/null 2>&1
```

Listing 3-12: autogen.sh: A temporary bootstrap script that executes the required Autotools

The automake --add-missing option copies the required missing utility scripts into the project, and the --copy option indicates that true copies should be

^{10.} See the Free Software Foundation's GNU Autoconf Manual at http://www.gnu.org/software/autoconf/manual/index.html.

made (otherwise, symbolic links are created that refer to the files where they're installed with the Automake package).¹¹

MISSING REQUIRED FILES IN AUTOCONF

When I first tried to execute autoreconf on the *configure.ac* file in Listing 3-11, I discovered a minor problem related to using Autoconf *without* Automake. When I ran the configure script, it failed with an error: configure: error: cannot find install-sh or install.sh ...

Autoconf is all about portability and, unfortunately, the Unix install utility is not as portable as it could be. From one platform to another, critical bits of installation functionality are just different enough to cause problems, so the Autotools provide a shell script called install-sh (deprecated name: install.sh). This script acts as a wrapper around the system's own install utility, masking important differences between various versions of install.

autoscan noticed that I'd used the install program in my *src/Makefile.in* template, so it generated an expansion of the AC_PROG_INSTALL macro. The problem is that configure couldn't find the install-sh wrapper script anywhere in my project.

I reasoned that the missing file was part of the Autoconf package, and it just needed to be installed. I also knew that autoreconf accepts a command-line option to install such missing files into a project directory. The --install option supported by autoreconf is designed to pass tool-specific options down to each of the tools that it calls in order to install missing files. However, when I tried that, I found that the file was still missing, because autoconf doesn't support an option to install missing files.¹

I could have manually copied install-sh from the Automake installation directory (usually /usr/(local/)share/automake-*), but looking for a more automated solution, I tried manually executing automake --add-missing --copy. This command generated a slew of warnings indicating that the project was not configured for Automake. However, I could now see that install-sh had been copied into my project root directory, and that's all I was after. Executing autoreconf --install didn't run automake because *configure.ac* was not configured for Automake.

Autoconf should ship with install-sh, since it provides a macro that requires it, but then autoconf would have to provide an --add-missing command-line option. Nevertheless, there is actually a quite obvious solution to this problem. The install-sh script is not really required by any code Autoconf generates. How could it be? Autoconf doesn't generate any makefile constructs—it only substitutes variables into your *Makefile.in* templates. Thus, there's really no reason for Autoconf to complain about a missing install-sh script.²

1. Worse still, the *GNU* Autoconf Manual that I was using at the time told me that "Autoconf comes with a copy of install-sh that you can use"—but it's really Automake and Libtool that come with copies of install-sh.

2. When I presented this problem on the Autoconf mailing list, I was told several times that autoconf has no business copying install-sh into a project directory, thus there is no install-missing-file functionality accessible from the autoconf command line. If this is indeed the case, then autoconf has no business complaining about the missing file, either!

^{11.} The automake --add-missing option copies in the missing required utility scripts, and the --copy option indicates that true copies should be made—otherwise, symbolic links are created to the files where the automake package has installed them. This isn't as bad as it sounds, because when make dist generates a distribution archive, it creates true copies in the image directory. Therefore, links work just fine, as long as you (the maintainer) don't move your work area to another host. Note that automake provides a --copy option, but autoreconf provides just the opposite: a --symlink option. Thus, if you execute automake --add-missing and you wish to actually copy the files, you should pass --copy as well. If you execute autoreconf --install, --copy will be assumed and passed to automake by autoreconf.

NOTE When make dist generates a distribution archive, it creates true copies in the image directory, so the use of symlinks causes no real problems, as long as you (the maintainer) don't move your work area to another host.

We don't need to see the warnings from automake, so I've redirected the stderr and stdout streams to /dev/null on the automake command line at ① in this script. In Chapter 5, we'll remove autogen.sh and simply run autoreconf --install, but for now, this will solve our missing file problems.

Updating Makefile.in

Let's execute autogen.sh and see what we end up with:

```
$ sh autogen.sh
$ ls -1p
autogen.sh
autom4te.cache/
① config.h.in
configure
configure.ac
② install-sh
Makefile.in
src/
$
```

We know from the file list at **0** that *config.h.in* has been created, so we know that autoreconf has executed autoheader. We also see the new install-sh script at **0** that was created when we executed automake in autogen.sh. Anything provided or generated by the Autotools should be copied into the archive directory so that it can be shipped with release tarballs. Therefore, we'll add cp commands for these two files to the \$(distdir) target in the top-level *Makefile.in* template. Note that we don't need to copy the autogen.sh script because it's purely a maintainer tool—users should never need to execute it from a tarball distribution.

Listing 3-13 illustrates the required changes to the \$(distdir) target in the top-level *Makefile.in* template.

```
% (distdir): FORCE
mkdir -p $(distdir)/src
cp $(srcdir)/configure.ac $(distdir)
cp $(srcdir)/config.h.in $(distdir)
cp $(srcdir)/config.h.in $(distdir)
cp $(srcdir)/install-sh $(distdir)
cp $(srcdir)/Makefile.in $(distdir)
cp $(srcdir)/src/Makefile.in $(distdir)/src
cp $(srcdir)/src/main.c $(distdir)/src
```

Listing 3-13: Makefile.in: Additional files needed in the distribution archive image directory

If you're beginning to think that this could become a maintenance problem, then you're right. I mentioned earlier that the \$(distdir) target was painful to maintain. Luckily, the distcheck target still exists and still works as designed. It would have caught this problem, because attempts to build from the tarball will fail without these additional files—and the check target certainly won't succeed if the build fails. When we discuss Automake in Chapter 5, we will clear up much of this maintenance mess.

Initialization and Package Information

Now let's turn our attention back to the contents of the *configure.ac* file in Listing 3-11. The first section contains Autoconf initialization macros. These are required for all projects. Let's consider each of these macros individually, because they're all important.

AC_PREREQ

The AC_PREREQ macro simply defines the earliest version of Autoconf that may be used to successfully process this *configure.ac* file:

```
AC_PREREQ(version)
```

The *GNU Autoconf Manual* indicates that AC_PREREQ is the only macro that may be used before AC_INIT. This is because it's good to ensure you're using a new enough version of Autoconf before you begin processing any other macros, which may be version dependent.

AC_INIT

The AC_INIT macro, as its name implies, initializes the Autoconf system. Here's its prototype, as defined in the *GNU Autoconf Manual*.¹²

AC_INIT(package, version, [bug-report], [tarname], [url])

It accepts up to five arguments (autoscan only generates a call with the first three): package, version, and optionally, bug-report, tarname, and url. The package argument is intended to be the name of the package. It will end up (in a canonical form) as the first part of the name of an Automake-generated release distribution tarball when you execute make dist.

NOTE Autoconf uses a normalized form of the package name in the tarball name, so you can use uppercase letters in the package name, if you wish. Automake-generated tarballs are named tarname-version.tar.gz by default, but tarname is set to a normalized form of the package name (lowercase, with all punctuation converted to underscores). Bear this in mind when you choose your package name and version string.

^{12.} The square brackets used in the macro definition prototypes within this book (as well as the *GNU Autoconf Manual*) indicate optional parameters, not Autoconf quotes.

The optional bug-report argument is usually set to an email address, but any text string is valid. An Autoconf substitution variable called <code>@PACKAGE_BUGREPORT@</code> is created for it, and that variable is also added to the *config.h.in* template as a C-preprocessor definition. The intent here is that you use the variable in your code to present an email address for bug reports at appropriate places—possibly when the user requests help or version information from your application.

While the version argument can be anything you like, there are a few commonly used OSS conventions that will make things a little easier for you. The most widely used convention is to pass in *major.minor* (e.g., 1.2). However, there's nothing that says you can't use *major.minor.revision*, and there's nothing wrong with this approach. None of the resulting VERSION variables (Autoconf, shell, or make) are parsed or analyzed anywhere—they're only used as placeholders for substituted text in various locations.¹³ So if you wish, you may even add nonnumeric text into this macro, such as *0.15.alpha1*, which is occasionally useful.¹⁴

NOTE The RPM package manager, on the other hand, does care what you put in the version string. For the sake of RPM, you may wish to limit the version string text to only alphanumeric characters and periods—no dashes or underscores.

The optional url argument should be the URL for your project website. It's shown in the help text displayed by configure --help.

Autoconf generates the substitution variables @PACKAGE_NAME@, @PACKAGE_VERSION@, @PACKAGE_TARNAME@, @PACKAGE_STRING@ (a stylized concatenation of the package name and version information), @PACKAGE_BUGREPORT@, and @PACKAGE_URL@ from the arguments to AC_INIT.

AC_CONFIG_SRCDIR

The AC_CONFIG_SRCDIR macro is a sanity check. Its purpose is to ensure that the generated configure script knows that the directory on which it is being executed is actually the project directory.

More specifically, configure needs to be able to locate itself, because it generates code that executes itself, possibly from a remote directory. There are myriad ways to inadvertently fool configure into finding some other configure script. For example, the user could accidentally provide an incorrect --srcdir argument to configure. The \$0 shell script parameter is unreliable, at best—it may contain the name of the shell, rather than that of the script, or it may be that configure was found in the system search path, so no path information was specified on the command line.

^{13.} As far as M4 is concerned, all data is text; thus M4 macro arguments, including package and version, are treated simply as strings. M4 doesn't attempt to interpret any of this text as numbers or other data types.

^{14.} A future version of Autoconf will support a public macro that allows lexicographical comparison of version strings, and certain internal constructs in current versions already use such functionality. Thus, it's good practice to form version strings that increase properly in a lexical fashion from version to version.

The configure script could try looking in the current or parent directories, but it still needs a way to verify that the configure script it locates is actually itself. Thus, AC_CONFIG_SRCDIR gives configure a significant hint that it's looking in the right place. Here's the prototype for AC_CONFIG_SRCDIR:

```
AC_CONFIG_SRCDIR(unique-file-in-source-dir)
```

The argument can be a path (relative to the project's configure script) to any source file you like. You should choose one that is unique to your project so as to minimize the possibility that configure is fooled into thinking some other project's configuration file is itself. I try to choose a file that sort of represents the project, such as a source file named for a feature that defines the project. That way, in case I ever decide to reorganize the source code, I'm not likely to lose it in a file rename. But it doesn't really matter, because both autoconf and configure will tell you and your users if it can't find this file.

The Instantiating Macros

Before we dive into the details of AC_CONFIG_HEADERS, I'd like to spend a little time on the file generation framework Autoconf provides. From a high-level perspective, there are four major things happening in *configure.ac*:

- Initialization
- Check request processing
- File instantiation request processing
- Generation of the configure script

We've covered initialization—there's not much to it, although there are a few more macros you should be aware of. Check out the *GNU Autoconf Manual* for more information—look up AC_COPYRIGHT, for an example. Now let's move on to file instantiation.

There are actually four so-called *instantiating macros*: AC_CONFIG_FILES, AC_CONFIG_HEADERS, AC_CONFIG_COMMANDS, and AC_CONFIG_LINKS. An instantiating macro accepts a list of tags or files; configure will generate these files from templates containing Autoconf substitution variables.

NOTE You might need to change the name of AC_CONFIG_HEADER (singular) to AC_CONFIG_HEADERS (plural) in your version of configure.scan. The singular version is the older name for this macro, and the older macro is less functional than the newer one.¹⁵

The four instantiating macros have an interesting common signature. The following prototype can be used to represent each of them, with appropriate text replacing the XXX portion of the macro name:

AC_CONFIG_XXXS(tag..., [commands], [init-cmds])

^{15.} This was a defect in autoscan that had not been fixed as of Autoconf version 2.61. However, version 2.62 of autoscan correctly generates a call to the newer, more functional AC_CONFIG_HEADERS.

For each of these four macros, the tag argument has the form OUT[:INLIST], where INLIST has the form IN0[:IN1:...:INn]. Often, you'll see a call to one of these macros with only a single argument, as in the three examples below (note that these examples represent macro *calls*, not *prototypes*, so the square brackets are actually Autoconf quotes, not indications of optional parameters):

```
AC_CONFIG_HEADERS([config.h])
```

In this example, *config.h* is the OUT portion of the above specification. The default value for INLIST is the OUT portion with *.in* appended to it. So, in other words, the above call is exactly equivalent to:

```
AC_CONFIG_HEADERS([config.h:config.h.in])
```

What this means is that config.status contains shell code that will generate *config.h* from *config.h.in*, substituting all Autoconf variables in the process. You may also provide a list of input files in the INLIST portion. In this case, the files in INLIST will be concatenated to form the resulting OUT file:

```
AC_CONFIG_HEADERS([config.h:cfg0:cfg1:cfg2])
```

Here, config.status will generate *config.h* by concatenating cfg0, cfg1, and cfg2 (in that order), after substituting all Autoconf variables. The *GNUAutoconf Manual* refers to this entire OUT[:INLIST] construct as a *tag*.

Why not just call it a *file*? Well, this parameter's primary purpose is to provide a sort of command-line target name—much like makefile targets. It can also be used as a filesystem name if the associated macro generates files, as is the case with AC_CONFIG_HEADERS, AC_CONFIG_FILES, and AC_CONFIG_LINKS.

But AC_CONFIG_COMMANDS is unique in that it doesn't generate any files. Instead, it runs arbitrary shell code, as specified by the user in the macro's arguments. Thus, rather than name this first parameter after a secondary function (the generation of files), the *GNU Autoconf Manual* refers to it more generally, according to its primary purpose—as a command-line *tag* that may be specified on the config.status command line, in this manner:

\$./config.status config.h

This config.status command line will regenerate the *config.h* file based on the macro call to AC_CONFIG_HEADERS in *configure.ac*. It will *only* regenerate *config.h*.

Enter ./config.status --help to see the other command-line options you can use when executing config.status:

```
$ ./config.status --help
'config.status' instantiates files from templates according to the
current configuration.
```

• Usage: ./config.status [OPTION]... [TAG]...

```
-h, --help
                     print this help, then exit
    -V, --version
                     print version number and configuration settings, then exit
    -q, --quiet, --silent
                     do not print progress messages
    -d, --debug
                     don't remove temporary files
        --recheck
                     update config.status by reconfiguring in the same
  conditions
        --file=FILE[:TEMPLATE]
ค
                     instantiate the configuration file FILE
        --header=FILE[:TEMPLATE]
                     instantiate the configuration header FILE
Onfiguration files:
   Makefile src/Makefile
Onfiguration headers:
   config.h
  Report bugs to <bug-autoconf@gnu.org>.
  $
```

Notice that config.status provides custom help about a project's config.status file. It lists configuration files ③ and configuration headers ④ that we can use as tags on the command line where the usage specifies [TAG]... at ①. In this case, config.status will only instantiate the specified objects. In the case of commands, it will execute the command set specified by the tag passed in the associated expansion of the AC_CONFIG_COMMANDS macro.

Each of these macros may be used multiple times in a *configure.ac* file. The results are cumulative, and we can use AC_CONFIG_FILES as many times as we need to in *configure.ac*. It is also important to note that config.status supports the --file= option (at **②**). When you call config.status with tags on the command line, the only tags you can use are those the help text lists as available configuration files, headers, links, and commands. When you execute config.status with the --file= option, you're telling config.status to generate a new file that's not already associated with any of the calls to the instantiating macros found in *configure.ac*. This new file is generated from an associated template using configuration options and check results determined by the last execution of configure. For example, I could execute config.status in this manner:

\$./config.status --file=extra:extra.in

NOTE The default template name is the filename with a .in suffix, so this call could have been made without using the :extra.in portion of the option. I added it here for clarity.

Let's return to the instantiating macro signature at the bottom of page 78. I've shown you that the tag... argument has a complex format, but the ellipsis indicates that it also represents multiple tags, separated by whitespace. The format you'll see in nearly all *configure.ac* files is shown in Listing 3-14.

```
...
AC_CONFIG_FILES([Makefile
src/Makefile
lib/Makefile
etc/proj.cfg])
```

Listing 3-14: Specifying multiple tags (files) in AC_CONFIG_FILES

Each entry here is one tag specification, which, if fully specified, would look like the call in Listing 3-15.

```
...
AC_CONFIG_FILES([Makefile:Makefile.in
src/Makefile:src/Makefile.in
lib/Makefile:lib/Makefile.in
etc/proj.cfg:etc/proj.cfg.in])
```

Listing 3-15: Fully specifying multiple tags in AC_CONFIG_FILES

Returning to the instantiating macro prototype, there are two optional arguments that you'll rarely see used in these macros: commands and init-cmds. The commands argument may be used to specify some arbitrary shell code that should be executed by config.status just before the files associated with the tags are generated. It is unusual for this feature to be used within the file-generating instantiating macros. You will almost always see the commands argument used with AC_CONFIG_COMMANDS, which generates no files by default, because a call to this macro is basically useless without commands to execute!¹⁶ In this case, the tag argument becomes a way of telling config.status to execute a specific set of shell commands.

The init-cmds argument initializes shell variables at the top of config.status with values available in *configure.ac* and configure. It's important to remember that all calls to instantiating macros share a common namespace along with config.status. Therefore, you should try to choose your shell variable names carefully so they are less likely to conflict with each other and with Autoconf-generated variables.

The old adage about the value of a picture versus an explanation holds true here, so let's try a little experiment. Create a test version of your *configure.ac* file that contains only the contents of Listing 3-16.

```
AC_INIT([test], [1.0])
AC_CONFIG_COMMANDS([abc],
[echo "Testing $mypkgname"],
[mypkgname=$PACKAGE_NAME])
AC_OUTPUT
```

Listing 3-16: Experiment #1-a simple configure.ac file

16. The truth is that we don't often use AC_CONFIG_COMMANDS.

Now execute autoreconf, configure, and config.status in various ways to see what happens:

```
$ autoreconf
• $ ./configure
  configure: creating ./config.status
  config.status: executing abc commands
  Testing test

9 $ ./config.status

  config.status: executing abc commands
  Testing test
❸ $ ./config.status --help
   'config.status' instantiates files from templates according to the current
  configuration.
  Usage: ./config.status [OPTIONS]... [FILE]...
  Configuration commands:
   abc
  Report bugs to <bug-autoconf@gnu.org>.
9 $ ./config.status abc
  config.status: executing abc commands
  Testing test
  $
```

As you can see at **0**, executing configure caused config.status to be executed with no command-line options. There are no checks specified in *configure.ac*, so manually executing config.status, as we did at **2**, has nearly the same effect. Querying config.status for help (as we did at **3**) indicates that abc is a valid tag; executing config.status with that tag (as we did at **4**) on the command line simply runs the associated commands.

In summary, the important points regarding the instantiating macros are as follows:

- The config.status script generates all files from templates.
- The configure script performs all checks and then executes config.status.
- When you execute config.status with no command-line options, it generates files based on the last set of check results.
- You can call config.status to execute file generation or command sets specified by any of the tags given in any of the instantiating macro calls.
- config.status may generate files not associated with any tags specified in *configure.ac*, in which case it will substitute variables based on the last set of checks performed.

AC_CONFIG_HEADERS

As you've no doubt concluded by now, the AC_CONFIG_HEADERS macro allows you to specify one or more header files that config.status should generate from template files. The format of a configuration header template is very specific. A short example is given in Listing 3-17.

```
/* Define as 1 if you have unistd.h. */
#undef HAVE_UNISTD_H
```

Listing 3-17: A short example of a header file template

You can place multiple statements like this in your header template, one per line. The comments are optional, of course. Let's try another experiment. Create a new *configure.ac* file like that shown in Listing 3-18.

```
AC_INIT([test], [1.0])
AC_CONFIG_HEADERS([config.h])
AC_CHECK_HEADERS([unistd.h foobar.h])
AC_OUTPUT
```

Listing 3-18: Experiment #2-a simple configure.ac file

Create a template header file called *config.h.in* that contains the two lines in Listing 3-19.

#undef HAVE_UNISTD_H
#undef HAVE_FOOBAR_H

Listing 3-19: Experiment #2 continued—a simple config.h.in file

Now execute the following commands:

```
$ autoconf
  $ ./configure
  checking for gcc... gcc
   . . .
O checking for unistd.h... yes
  checking for unistd.h... (cached) yes
  checking foobar.h usability... no
  checking foobar.h presence... no
Ochecking for foobar.h... no
  configure: creating ./config.status
⑤ config.status: creating config.h
  $ cat config.h
  /* config.h. Generated from ... */
  #define HAVE UNISTD H 1
④ /* #undef HAVE FOOBAR H */
  $
```

You can see at O that config.status generated a *config.h* file from the simple *config.h.in* template we wrote. The contents of this header file are based on the checks executed by configure. Since the shell code generated by AC_CHECK_HEADERS([unistd.h foobar.h]) was able to locate a *unistd.h* header file (O) in the system include directory, the corresponding #undef statement was converted into a #define statement. Of course, no *foobar.h* header was found in the system include directory, as you can also see by the output of configure at O; therefore, its definition was left commented out in the template, as shown at O.

Thus, you may add the sort of code shown in Listing 3-20 to appropriate C-language source files in your project.

```
#if HAVE_CONFIG_H
# include <config.h>
#endif
#if HAVE_UNISTD_H
# include <unistd.h>
#endif
#if HAVE_FOOBAR_H
# include <foobar.h>
#endif
```

Listing 3-20: Using generated CPP definitions in a C-language source file

Using autoheader to Generate an Include File Template

Manually maintaining a *config.h.in* template is more trouble than necessary. The format of *config.h.in* is very strict—for example, you can't have any leading or trailing whitespace on the #undef lines. Besides that, most of the information you need from *config.h.in* is available in *configure.ac*.

Fortunately, the autoheader utility will generate a properly formatted header file template for you based on the contents of *configure.ac*, so you don't often need to write *config.h.in* templates. Let's return to the command prompt for a final experiment. This one is easy—just delete your *config.h.in* template and then run autoheader and autoconf:

```
$ rm config.h.in
$ autoheader
$ autoconf
$ ./configure
checking for gcc... gcc
...
checking for unistd.h... yes
checking for unistd.h... (cached) yes
checking foobar.h usability... no
checking foobar.h presence... no
checking for foobar.h... no
```

```
configure: creating ./config.status
config.status: creating config.h
$
$ cat config.h
/* config.h. Generated from config.h.in... */
/* config.h.in. Generated from configure.ac... */
. . .
/* Define to 1 if you have... */
/* #undef HAVE FOOBAR H */
/* Define to 1 if you have... */
#define HAVE UNISTD H 1
/* Define to the address where bug... */
#define PACKAGE_BUGREPORT ""
/* Define to the full name of this package. */
#define PACKAGE NAME "test"
/* Define to the full name and version... */
#define PACKAGE STRING "test 1.0"
/* Define to the one symbol short name... */
#define PACKAGE TARNAME "test"
/* Define to the version... */
#define PACKAGE VERSION "1.0"
/* Define to 1 if you have the ANSI C... */
#define STDC HEADERS 1
$
```

NOTE Again, I encourage you to use autoreconf, which will automatically run autoheader if it notices an expansion of AC_CONFIG_HEADERS in configure.ac.

As you can see by the output of the cat command at $\mathbf{0}$, an entire set of preprocessor definitions was derived from *configure.ac* by autoheader.

Listing 3-21 shows a much more realistic example of using a generated *config.h* file to increase the portability of your project source code. In this example, the AC_CONFIG_HEADERS macro call indicates that *config.h* should be generated, and the call to AC_CHECK_HEADERS will cause autoheader to insert a definition into *config.h*.

```
AC_INIT([test], [1.0])
AC_CONFIG_HEADERS([config.h])
AC_CHECK_HEADERS([dlfcn.h])
AC_OUTPUT
```

Listing 3-21: A more realistic example of using AC_CONFIG_HEADERS

The *config.h* file is intended to be included in your source code in locations where you might wish to test a configured option in the code itself using the C preprocessor. This file should be included first in source files so it can influence the inclusion of system header files later in the source.

NOTE The config.h.in template that autoheader generates doesn't contain an include-guard construct, so you need to be careful that it's not included more than once in a source file.

It's often the case that every .c file in a project needs to include *config.h*. In this case, it might behoove you to include *config.h* at the top of an internal project header file that's included by all the source files in your project. You can (and probably should) also add an include-guard construct to this internal header file to protect against including it more than once.

Don't make the mistake of including *config.h* in a public header file if your project installs libraries and header files as part of your product set. For more detailed information on this topic, refer to "Item 1: Keeping Private Details out of Public Interfaces" on page 272.

Using the *configure.ac* file from Listing 3-21, the generated configure script will create a *config.h* header file with appropriate definitions for determining, at compile time, whether or not the current system provides the dlfcn interface. To complete the portability check, you can add the code from Listing 3-22 to a source file in your project that uses dynamic loader functionality.

```
#if HAVE_CONFIG_H
# include <config.h>
#endif

# if HAVE_DLFCN_H
# include <dlfcn.h>
#else
# error Sorry, this code requires dlfcn.h.
#endif
...
#if HAVE_DLFCN_H
handle = dlopen("/usr/lib/libwhatever.so", RTLD_NOW);
#endif
...
```

Listing 3-22: A sample source file that checks for dynamic loader functionality

If you already had code that included dlfcn.h, autoscan would have generated a line in *configure.ac* to call AC_CHECK_HEADERS with an argument list containing dlfcn.h as one of the header files to be checked. Your job as maintainer is to add the conditional statements at **0** and **2** to your source code around the existing inclusions of the dlfcn.h header file and around calls to the dlfcninterface functions. This is the crux of Autoconf-provided portability.

Your project might prefer dynamic loader functionality, but could get along without it if necessary. It's also possible that your project requires a dynamic loader, in which case your build should terminate with an error (as this code does) if the key functionality is missing. Often, this is an acceptable stopgap until someone comes along and adds support to the source code for a more system-specific dynamic loader service.

NOTE If you have to bail out with an error, it's best to do so at configuration time rather than at compile time. The general rule of thumb is to bail out as early as possible.

One obvious flaw in this source code is that *config.h* is only included if HAVE_CONFIG_H is defined in your compilation environment. You must define HAVE_CONFIG_H manually on your compiler command lines if you're writing your own makefiles. Automake does this for you in generated *Makefile.in* templates.

HAVE_CONFIG_H is part of a string of definitions passed on the compiler command line in the Autoconf substitution variable @DEFS@. Before autoheader and AC_CONFIG_HEADERS functionality existed, Automake added all of the compiler configuration macros to the @DEFS@ variable. You can still use this method if you don't use AC_CONFIG_HEADERS in *configure.ac*, but it's not recommended mainly because a large number of definitions make for very long compiler command lines.

Back to Remote Builds for a Moment

As we wrap up this chapter, you'll notice that we've come full circle. We started out covering some preliminary information before we discussed how to add remote builds to Jupiter. Now we'll return to this topic for a moment, because I haven't yet covered how to get the C preprocessor to properly locate a generated *config.h* file.

Since this file is generated from a template, it will be at the same relative position in the build directory structure as its counterpart template file, *config.h.in*, is in the source directory structure. The template is located in the top-level *source* directory (unless you chose to put it elsewhere), so the generated file will be in the top-level *build* directory. Well, that's easy enough—it's always one level up from the generated *src/Makefile*.

Before we draw any conclusions then about header file locations, let's consider where header files might appear in a project. We might generate them in the current build directory, as part of the build process. We might also add internal header files to the current source directory. We know we have a *config.h* file in the top-level build directory. Finally, we might also create a top-level *include* directory for library interface header files our package provides. What is the order of priority for these various *include* directories?

The order in which we place *include directives* (-Ipath options) on the compiler command line is the order in which they will be searched, so the order should be based on which files are most relevant to the source file currently being compiled. Thus, the compiler command line should include -Ipath directives for the current build directory (.) first, followed by the source directory [\$(srcdir)], then the top-level build directory (..), and finally, our project's *include* directory, if it has one. We impose this ordering by adding -Ipath options to the compiler command line, as shown in Listing 3-23.

```
...
jupiter: main.c
    $(CC) -I. -I$(srcdir) -I.. $(CPPFLAGS) $(CFLAGS) -o $@ main.c
...
```

Listing 3-23: src/Makefile.in: Adding proper compiler include directives

Now that we know this, we need to add another rule of thumb for remote builds to the list we created on page 69:

• Add preprocessor commands for the current build directory, the associated source directory, and the top-level build directories, in that order.

Summary

In this chapter, we covered just about all the major features of a fully functional GNU project build system, including writing a *configure.ac* file, from which Autoconf generates a fully functional configure script. We've also covered adding remote build functionality to makefiles with VPATH statements.

So what else is there? Plenty! In the next chapter, I'll continue to show you how you can use Autoconf to test system features and functionality before your users run make. We'll also continue enhancing the configuration script so that when we're done, users will have more options and understand exactly how our package will be built on their systems.